EDM SLOPE STAKING
WITH THE HP-41CV/CX



9+25.000 107.88
9+50.000 107.50
9+75.000 107.13
10+00.000 106.75
10+25.000 106.38
200 FT V.C.
11+50.000 PVI 104.50
POVC 105.88
11+04.545 G=0% 105.59

prompt: STA?

keystrokes:

2 5 1

output:

10+75.000 105.71
11+00.000 105.59



Software by D'Zign
P.O. BOX 1370 • PACIFICA, CA 94044

NOTICE

No express or implied warranty is made by D'Zign Land Survey & Development or the author with regard to the procedures and program material offered or their merchantability or their fitness for any particular purpose. The keystroke procedures and program material are made available solely on an "as-is" basis, and the entire risk as to their quality and performance is with the user. Should the procedures or program material prove defective, the user (and not D'Zign Land Survey & Development nor any other party) shall bear any and all cost of all necessary correction and all incidental or consequential damages. D'Zign Land Survey & Development and/or the author shall not be liable for any incidental or consequential damages in connection with or arising out of the furnishing, use, or performance of the keystroke procedures or program material.

EDM SLOPE STAKING

WITH THE HP-41CV/CX

Copyright © 1987 by Ted J. Kerber

All rights reserved. No part of this work covered by the copyright hereon may be reproduced or used in any form or by any means - graphic, electronic, or mechanical, including photocopying, recording, taping, or information storage and retrieval systems - without written permission of the author.

published by D'Zign Land Survey & Development
Pacifica, California 1987

ISBN 9616846-8-2

Setting slope stakes is a time-consuming process, sometimes very frustrating, and always necessary. We've all wished that there was an easier way to get the job done.

As the prices on electronic distance devices kept getting lower, more and more surveyors made the switch. The use of a top-mount EDM or any type of total station also allows the use of programmed calculators to do the calculations.

The program in this booklet will do the calculations; all you have to enter are the horizontal angle, zenith angle, slope distance and rod reading. It is best (when possible) to set an even foot rod height as a constant.



Other than the R/S button, you will only need to remember three keys on the board, so you won't need a special overlay.

C is used for the **cut** factor and F is used for the **fill** factor The other key used is A, which is stroked after input of the shot data. The program is self-prompting for all needed input, so you won't need to carry the book around either.

The cut and fill factors mentioned above are only entered once, and allow you to change from a cut section to a fill section with just one keystroke. These functions are described in more detail in the user instructions.

If it becomes necessary to vary either of them (starting to stake to a bench, instead of the original hinge point, etc.) you may simply store a new cut hinge point elevation in register 27, or fill constant in register 28. The factors are the elevation of the hinge point at the instrument station.

Introduction

When the station that is occupied by the instrument is within a vertical curve, the elevation that is input as profile grade is the elevation of the **vertical tangent** at the instrument station.



The slope ratio and the half-width are also carried as constants.

These are originally displayed as "zero" values, and may be changed at any shot to fit the slope you are staking. This feature is useful in cases such as a change from cut to fill at the station shot, or if you want to "flatten" the slopes as you approach a daylight area.

After each shot, the distance to the catch point (at the same elevation as the shot) is displayed, so that you can evaluate the location of the catch point for the next shot. When the distance is within acceptable (±0.2', usually) accuracy, stroking R/S outputs the cut or fill information for writing the marker stake.

A routine has also been included which calculates the angle to turn and the distance to measure to set the reference stake. The amount of offset is input by the user and the angle and distance are calculated. A shot taken on the reference hub after it is set will output the elevation of the RP and the plus or minus to the slope stake.

This booklet has been published in two versions, one with programmed magnetic cards and one without. If your version includes the cards, simply input the program and try the practice keystroke examples. If your version does not include the cards, input the program steps as shown in the program listings.

It is a fairly long program, but keep in mind all that we are asking it to do, and think ahead to the amount of time and work it is going to save.

To begin, the program "**SS**" should be in program memory, and the calculator sized at **030** (or larger). It is sometimes convenient to have additional storage registers available for "scratch" storage.

Bring the program to the top of program memory and initialize it by keystroking XEQ ALPHA S S ALPHA. This clears the registers, resets the status of the flags, and calls up the first prompt:

INST STA?

1. Input the station which (or opposite which) is occupied by the instrument. If this station is within a curve on the alignment, stroke CHS before stroking R/S

OFFSET?

2. If the instrument is on centerline, input 0. If the instrument is at an offset distance from centerline, input the offset distance. If the offset is left of centerline, stroke CHS before R/S

HI?

3. Input the elevation of the height of instrument. R/S

BKSITE STA?

4. Input the station at the backsight. If the station is within a curve (and the instrument station was not), first stroke CHS, then R/S

OFFSET?

5. Input 0 if the sight is on centerline, or input the offset distance. If the offset of the point is to the left of centerline, stroke CHS before R/S

B.C. STA?

6. This prompt will appear after the instrument station input, if the instrument

is on a curve in the alignment, or after input of the backsight information.

6a If neither the instrument nor backsight were on a curve, but there is a curve in the alignment within the work area, input the station at the beginning of the curve.

OR

6b If there is no curve, input a station which is beyond the work area, and CHS. This will act as an auto-stop at that station. If there is a curve, the prompts marked * will appear.

R/S

RADIUS*

7. Input the length of the radius of the curve

R/S

DELTA?*

8. Input the central angle of the curve. If the curve is a curve to the left, CHS before stroking

R/S

PROFILE EL?

9. Input the **finished grade** elevation at the instrument station. If the instrument is at a station which is located within a vertical curve, input the elevation of the **tangent profile grade.**

R/S

C. FACTOR?

10. Input the "cut factor". This is the difference in elevation between the finished (profile) grade and the hinge-point of the cut section. If negative, CHS, then stroke

R/S

F. FACT?

11. Input the "fill factor". This is the difference between profile grade and the hinge-point in a fill section. If negative, CHS, and stroke

R/S

% GRADE?

12. Input the percent of grade for the vertical alignment. If negative, CHS. Stroke R/S

BVC STA?

13a Input the station at the beginning of the vertical curve. In the case of a grade break, instead of a curve, input the station at the grade break

OR

13b If there is no vertical curve input a station beyond the work area and stroke CHS to cancel the next two prompts. R/S

LENGTH?

14. Input the length of the vertical curve. If this is a grade break instead of a curve, input 0 R/S

GRADE OUT?

15. Input the percent of grade leaving the vertical curve. If negative, CHS before stroking R/S

INPUT SHOT

The first time this prompt appears, or at any time when you want to change the hinge point elevation (from cut to fill or from fill to cut) select the appropriate factor

16a If staking a cut, stroke C

OR

16b If staking a fill, stroke F

INPUT SHOT

17. Input the horizontal angle ENTER↓

Input the zenith angle ENTER↓

Input the measured slope distance ENTER↓

Input the rod reading A

The next two prompts are for the "half-width" and the slope ratio. Since they both vary, the value about to be used for the calculation is displayed, and may be changed at this time by input of a new value, prior to stroking R/S.

W/2=0.00

18. Input the correct half-width value, if it is different than the value which is displayed

R/S

SR=0:1

19. Input the correct slope ratio value, if different from the value displayed

R/S

At this point the display will show the distance to the actual catch point. If the distance is considered to be within tolerance, R/S. If it is not, have the rodman move to a new position (based on the display) and return to step 17 with input of the data from the new shot.

Once you have reached the catch point, stroking R/S will display the information for making out the slope stake, and begin the routine for setting the reference stake. This latter routine begins with the prompt

RP AT?

20. Input the distance from the slope stake to the reference stake. If it is left of centerline, CHS, then

R/S

Output will be the distance to the reference point and the horizontal angle to turn to set it.

INPUT SHOT

21. After setting the reference point, take a shot on it and input it in the same manner as in step 17.

Output will be the elevation of the reference point and the difference in elevation to the slope stake.


26'-0"
30'-0"
8'-6"
17'-6"
17'-6"
8'-6"
4'-0"
4% +
+0.49'
+2%
2% +
-1.51'
+4%
TYPICAL FILL SECTION
TYPICAL CUT SECTION
R = 800.00'
Δ = 8°57'09"
L = 125.00'
T = 62.63'
C = 124.87'
13+75 E.C.
12+50 B.C.
Backsight is at station 13+75 E.C.
top of embankment
flowline of ditch
Instrument is 7.00' left of station 10+00
H.I. 113.62
10+50 PVC 106.00
11+50 PVI 104.50
12+50 PVT 108.50
-1.5%
+4.0%
115
110
105
10
11
12
13
200' V.C.

Keystroke Examples

Before you begin input of the alignment information, take a moment to calculate the cut and fill factors. This is simply the difference in elevation between the finish profile grade and the hinge point you are staking to. If for instance, you were staking to the back of a bench in a cut section, this could be a difference of 20 or more feet.

We are going to use the roadway details shown on the opposite page for the keystroke examples, so we won't have that type of condition. In the typical cut section we are going to be staking to the flowline of the ditch.

The **factor** will be the sum of the differences. We have 17.5 feet at -2%, +0.5' for the curb, and 8.5' at +4% on both sides; (17.5)(-.02) + 0.5 + (8.5)(.04) = +0.49. This is the **fill factor.** For a cut, we have the added ditch, 4' @ 2:1, which is -2', so the **cut factor** will be -1.51'.

Let's start with the keystrokes for input of the basic information about our setup and the alignments of our roadway. The calculator should be sized at **030**, and we initialize the program by stroking XEQ ALPHA S S ALPHA

prompt: **INST STA?**

keystrokes:

1 0 0 0 R/S

prompt: **OFFSET?**

keystrokes:

7 CHS R/S

prompt: **HI=?**

keystrokes:

1 1 3 . 6 2 R/S

prompt: **BKSITE STA?**

keystrokes:

1 3 7 5 CHS R/S

(it's on a curve)

prompt: **OFFSET?**

keystrokes:

0 R/S

prompt: **B.C. STA?**

keystrokes

1 2 5 0 R/S

prompt: **RADIUS?**

keystrokes:

8 0 0 R/S

prompt: **DELTA?**

keystrokes:

8 . 5 7 0 9 R/S

prompt: **PROFILE EL?**

keystrokes:

1 0 6 . 7 5 R/S

prompt: **C. FACT?**

keystrokes:

1 . 5 1 CHS R/S

prompt: **F. FACT?**

keystrokes:

. 4 9 R/S

prompt: **GRADE?**

keystrokes:

1 . 5 CHS R/S

prompt: **BVC STA?**

keystrokes:

1 0 5 0 R/S

prompt: **LENGTH?**

keystrokes:

2 0 0 R/S

prompt: **GRADE OUT?**

keystrokes:

4 R/S

prompt: **INPUT SHOT**

Prior to the first shot input, we need to say whether it's a cut or fill section. We'll assume that the first shot will be a fill, and stroke F

prompt: **INPUT SHOT**

Let's try a shot using the following data:

horiz. angle = 13°10'20"
zenith angle = 92°05'15"
slope distance = 187.40'
the rod reading is 5.00'

keystrokes:

1 3 . 1 0 2 ENTER↑

9 2 . 0 5 1 5 ENTER↑

1 8 7 . 4 ENTER↑

5 A

prompt: **W/2= 0.00**

keystrokes:

2 6 R/S

prompt: **SR= 0:1**

keystrokes:

3 R/S

display: 2.55

You need to have the rodman go about $2\frac{1}{2}$ feet to the right. We take a shot there, with the following data:

horiz. angle = 13°57'00"
zenith angle = 92°42'00"
slope distance = 187.90'
the rod reading is 5.00'

Keystroke Examples

Input this data the same as we did the first shot. This time the prompts for the half-width and the slope ratio will already show us the right information, so we can just continue at that point.

keystrokes:



prompt: **W/2= 26.00**

keystroke: R/S

prompt: **SR= 3:1**

keystroke: R/S

display: -0.94

The ground slope has made a difference, and you need to go left about a foot, at the same elevation, for a catch point. On this shot we get:

horiz. angle = 13°30'25"
zenith angle = 92°27'00"
slope distance = 187.70'
the rod reading is 5.00'

keystrokes:



prompt: **W/2= 26.00'**

keystroke: R/S

prompt: **SR= 3:1**

keystroke: R/S

display: 0.08

This is close enough to use for the catch point. Stroke R/S

output:

```
* FILL 6.3
*  AT 18.8
* 11+80.20
* AT 44.90 RT
* ELEV = 100.60
* GR = 106.87
```

prompt: **RP AT?**

Let's set the reference point at 15'

keystrokes: 1 5 R/S

* R/S will be needed if used without printer

output:

```
* DIST=192.22
* ∠=17.4827
*
```

prompt: **INPUT SHOT**

Turn the angle shown and set the reference point at the distance given. After the point is set, take a shot on it to get the elevation. We'll use the following data:

horiz. angle = 17°48'30"
zenith angle = 93°06'15"
slope distance = 192.20'
the rod reading is 5.00'

keystrokes:



output:

```
* RP EL=98.21
* +2.38 TO S.S.
*
```

prompt: **INPUT SHOT**

Let's try one on the cut side. First, change the factor by stroking C

We'll use the following data:

horiz. angle = 348°23'00"
zenith angle = 89°19'50"
slope distance = 201.90'
the rod reading is 5.00'

keystrokes:



prompt: **W/2= 26.00**

keystrokes:

prompt: **SR= 3.00**

keystrokes:

display: 0.29

Let's go a little to the right and try another shot. We get:

horiz. angle = 348°25'40"
zenith angle = 89°17'20"
slope distance = 202.30'
the rod reading is 5.00'

keystrokes:



Keystroke Examples

prompt: **W/2= 30.00**

keystroke: R/S

prompt: **SR= 1.50:0**

keystroke: R/S

display: -0.01

keystroke: R/S

output:

```
CUT 5.8
* AT 8.7
*11+99.79
*AT 38.69 LT
*ELEV = 111.13
*GR = 105.33
```

prompt: **RP AT?**

keystrokes: 1 0 CHS R/S

(we're going left)

output:

```
DIST=204.09
* ∠=345.3914
*
```

prompt: **INPUT SHOT**

After setting the RP, we get a shot of:

horiz. angle = 345°39'00"
zenith angle = 87°32'00"
slope distance = 204.10'
the rod reading is 5.00'

keystrokes:

3 4 5 . 3 9 ENTER↑
8 7 . 3 2 ENTER↑
2 0 4 . 1 ENTER↑
5 A

output:

```
* RP EL=117.40
* -6.27 TO S.S.
```

prompt: **INPUT SHOT**

That is really all there is to it; the keystroke examples are typical of what you will encounter in the field.

There is one "rule of thumb" to learn: When the distance from the catch point is displayed after a shot, the sign tells you which way to go, at the same elevation, to get to the catch point.

A negative distance means go further left EXCEPT IN A CUT THAT IS RIGHT OF CENTERLINE.

A positive distance means go further right EXCEPT IN A CUT THAT IS RIGHT OF CENTERLINE.

We felt, when we were finishing this program, that the user would rather remember one "rule of thumb" than type in the additional steps that would eliminate this exception. It is already a long program.

* R/S will be needed if used without printer

The shots listed below will give you some additional practice with the keystrokes. The shots are to the catch points, still using the same alignment. Remember to change the cut or fill factor (if necessary) before input of the shot. Forgetting to do so will result in an error at the catch point.

1. **Cut section**, set the RP @ 15'. Use a zenith angle of 88°12'20" for the RP shot.

   horiz. angle = 354°40'50"
   zenith angle = 88°29'25"
   slope distance = 400.70'
   the rod reading is 5.00'

```
* CUT 6.4                  DIST=403.87  *
*  AT 9.6                  ∠=352.3547   *
* 13+94.69                              *
* AT 39.57 LT              RP EL=121.27 *
* ELEV = 119.18            -2.09 TO S.S.*
* GR = 112.78                           *
```

2. **Fill section,** set the RP @ 15'. Use a zenith angle of 89°40'30" for the RP shot.

   horiz. angle = 6°13'55"
   zenith angle = 89°25'40"
   slope distance = 477.50'
   the rod reading is 5.00'

```
* FILL 4.8                  DIST=477.67  *
*  AT 14.4                 ∠=8.0153     *
* 14+79.94                              *
* AT 40.40 RT              RP EL=111.33 *
* ELEV = 113.39            +2.06 TO S.S.*
* GR = 118.19                           *
```

3. **Cut section**, set the RP @ 15'. Use a zenith angle of 87°52'10" for the RP shot.

   horiz. angle = 351°23'30"
   zenith angle = 88°21'00"
   slope distance = 315.00'
   the rod reading is 5.00'

```
* CUT 8.3                   DIST=317.39  *
*  AT 12.5                 ∠=348.4248   *
* 13+09.99                              *
* AT 42.54 LT              RP EL=120.42 *
* ELEV = 117.69            -2.73 TO S.S.*
* GR = 109.39                           *
```

The two most likely sources of error when using this program are forgetting to change the factor when going from cut to fill (or fill to cut), and not checking that the half-width and slope ratio are correct for the shot being taken.

* R/S will be needed if used without printer

Program Listings

The program "**SS**" (**S**lope **S**take) contains 1,079 bytes of programming and occupies 154 registers. It may be stored onto 10 tracks of magnetic cards. If your version of this booklet includes the programmed cards, simply input the program from the cards.

If you are going to type the program into memory, the complete listing of the program and its subroutines is shown below. Care should be taken to insure that the program steps are correct, by proof-reading the program after input.

```
01♦LBL "SS"
02 SF 01
03 CF 02
04 CF 03
05 CF 04
06 CF 05
07 CF 06
08 CF 08
09 SF 21
10 SF 27
11 CF 29
12 CLRG
13 "INST STA?"
14 PROMPT
15 STO 05
16 "OFFSET?"
17 PROMPT
18 STO 06
19 RCL 05
20 X<0?
21 XEQ 06
22 "HI=?"
23 PROMPT
24 STO 11
25 "BKSITE STA?"
26 PROMPT
27 STO 03
28 "OFFSET?"
29 PROMPT
30 STO 04
31 RCL 03
32 X<0?
33 XEQ 07
34 RCL 03
35 RCL 04
36 RCL 06
37 -
38 X<>Y
39 RCL 05
40 -
41 R-P
42 CLX
43 X<>Y
44 X<0?
45 360
46 +
47 STO 07
48 FS? 01
49 XEQ 00
50 "PROFILE EL?"
51 PROMPT
52 STO 20
53 ENTER↑
54 ENTER↑
55 "C. FACT?"
56 PROMPT
57 +
58 STO 27
59 RDN
60 "F. FACT?"
61 PROMPT
62 +
63 STO 28
64 "% GRADE?"
65 PROMPT
66 1 E2
67 /
68 STO 10
69♦LBL 09
70 SF 04
71 "BVC STA?"
72 PROMPT
73 X<0?
74 SF 02
75 X<0?
76 CHS
77 FS? 02
78 GTO 10
79 STO 16
80 "LENGTH?"
81 PROMPT
82 STO 17
83 "GRADE OUT?"
84 PROMPT
85 1 E2
86 /
87 STO 09
88 GTO 10
89♦LBL 06
90 CHS
91 STO 05
92 XEQ 00
93 XEQ 01
94 RTN
95♦LBL 07
96 CHS
97 STO 03
98 FS? 01
99 XEQ 00
100 XEQ 05
101 RTN
102♦LBL A
103 GTO A
104♦LBL 00
105 CF 01
106 "B.C. STA?"
107 PROMPT
108 X<0?
109 SF 01
110 X<0?
111 CHS
112 STO 15
113 FS? 01
114 RTN
115 "RADIUS?"
116 PROMPT
117 STO 14
118 "DELTA?"
```

```
119 PROMPT
120 HR
121 X<0?
122 SF 05
123 STO 13
124 D-R
125 RCL 14
126 *
127 ABS
128 RCL 15
129 +
130 STO 24
131 RCL 13
132 ABS
133 2
134 /
135 TAN
136 RCL 14
137 *
138 STO 26
139 -
140 STO 25
141 RCL 15
142 ST+ 26
143 RTN
144♦LBL 01
145 RDN
146 CLX
147 RCL 14
148 PI
149 *
150 RCL 15
151 RCL 05
152 XEQ 03
153 RCL 06
154 XEQ 04
155 STO 06
156 X<>Y
157 RCL 15
158 +
159 STO 05
160 RTN
161♦LBL 05
162 RCL 14
163 PI
164 *
165 RCL 15
166 RCL 03
167 XEQ 03
168 RCL 04
169 XEQ 04
170 STO 04
171 X<>Y
172 RCL 15
173 +
174 STO 03
175 RTN
176♦LBL 03
177 -
178 CHS
179 180
180 *
181 X<>Y
182 /
183 RTN
184♦LBL 04
185 FC? 05
186 CHS
187 RCL 14
188 +
189 P-R
190 CHS
191 RCL 14
192 +
193 FS? 05
194 CHS
195 RTN
196♦LBL 17
197 FIX 2
198 CLA
199 "RP EL="
200 ARCL X
201 AVIEW
202 RCL 12
203 -
204 CHS
205 CLA
206 X>0?
207 "+"
208 ARCL X
209 "├ TO S.S."
210 AVIEW
211 RTN
212♦LBL A
213 SF 03
214♦LBL B
215 CHS
216 STO 01
217 RDN
218 X<>Y
219 HR
220 X<>Y
221 P-R
222 RCL 11
223 +
224 RCL 01
225 +
226 FS? 06
227 XEQ 17
228 FS?C 06
229 GTO 10
230 STO 12
231 RDN
232 X<>Y
233 HR
234 RCL 07
235 +
236 X<>Y
237 P-R
238 RCL 05
239 +
240 STO 18
241 X<>Y
242 RCL 06
243 +
244 STO 19
245 X<>Y
246 RCL 24
247 X<=Y?
248 GTO 08
249 CLX
250 RCL 15
251 FS? 01
252 GTO 11
253 X>Y?
254 GTO 11
255 -
256 X<>Y
257 FS? 05
258 CHS
259 RCL 14
260 -
261 CHS
262 R-P
263 CHS
264 RCL 14
265 +
266 FS? 05
267 CHS
268 STO 19
269 X<>Y
270 D-R
271 RCL 14
272 *
273 RCL 15
274 +
275 STO 18
276 RCL 24
277 X<=Y?
278 GTO 08
279 RDN
280 X<>Y
281 FS? 05
282 CHS
283 X<>Y
284 ENTER↑
285 GTO 11
286 RTN
```

```
287♦LBL A
288 GTO A
289♦LBL 02
290 RCL 08
291 "W/2= "
292 ARCL X
293 PROMPT
294 STO 08
295 SF 08
296 RCL 29
297 ENTER↑
298 INT
299 X=Y?
300 CF 08
301 RDN
302 FIX 0
303 FS? 08
304 FIX 2
305 "SR= "
306 ARCL X
307 "⊢:1"
308 PROMPT
309 FIX 2
310 STO 29
311 RCL 23
312 RCL 12
313 -
314 STO 02
315 ABS
316 RCL 29
317 *
318 RCL 08
319 +
320 STO 22
321 RCL 21
322 X<0?
323 CHS
324 -
325 CHS
326 STOP
327 FIX 1
328 CLA
329 "FILL "
330 RCL 02
331 X<0?
332 "CUT "
333 ABS
334 ARCL X
335 AVIEW
336 RCL 29
337 *
338 " AT "
339 ARCL X
340 AVIEW
341 CF 03
342 RCL 21
343 RCL 18
344 ENTER↑
345 GTO 11
346 RTN
347♦LBL 11
348 RDN
349 XEQ 12
350 CLX
351 X<>Y
352 RND
353 STO 21
354 X>Y?
355 GTO 13
356 CHS
357 "AT "
358 ARCL X
359 "⊢ LT"
360♦LBL 15
361 FC? 03
362 AVIEW
363 "ELEV = "
364 ARCL 12
365 FC? 03
366 AVIEW
367 XEQ 14
368 FS? 03
369 GTO 02
370 GTO 10
371 RTN
372♦LBL 10
373 "INPUT SHOT"
374 PROMPT
375 RTN
376♦LBL 13
377 "AT "
378 ARCL X
379 "⊢ RT"
380 GTO 15
381♦LBL 12
382 CLA
383 FIX 0
384 STO Y
385 1 E2
386 /
387 ENTER↑
388 INT
389 ARCL X
390 -
391 "⊢+"
392 FIX 2
393 1 E2
394 *
395 10
396 X>Y?
397 "⊢0"
398 ARCL Y
399 R↑
400 R↑
401 FC? 03
402 AVIEW
403 RCL 19
404 ENTER↑
405 ENTER↑
406 RTN
407♦LBL A
408 GTO A
409♦LBL 14
410 RCL 18
411 FS? 04
412 GTO 21
413♦LBL 26
414 RCL 18
415 RCL 05
416 -
417 RCL 10
418 *
419 RCL 20
420♦LBL 24
421 +
422 STO 23
423 "GR = "
424 ARCL X
425 FC? 03
426 AVIEW
427 FS? 03
428 GTO 02
429 SF 06
430 "RP AT?"
431 PROMPT
432 RCL 19
433 +
434 RCL 06
435 -
436 RCL 18
437 RCL 24
438 X<=Y?
439 GTO 20
440 CLX
441 RCL 15
442 X<=Y?
443 XEQ 19
444 RDN
445 RCL 05
446 -
447♦LBL 16
448 R-P
449♦LBL 18
450 FIX 2
451 "DIST="
452 ARCL X
453 AVIEW
454 FIX 4
```

```
455 CLX
456 X<>Y
457 RCL 07
458 -
459 X<=Y?
460 360
461 +
462 HMS
463 "∠="
464 ARCL X
465 AVIEW
466 GTO 10
467 RTN
468♦LBL 19
469 -
470 RCL 14
471 /
472 R-D
473 STO 03
474 COS
475 X<>Y
476 FC? 05
477 CHS
478 RCL 14
479 +
480 *
481 STO 04
482 RCL 03
483 TAN
484 *
485 RCL 15
486 RCL 05
487 -
488 +
489 RCL 04
490 RCL 14
491 -
492 FC? 05
493 CHS
494 X<>Y
495 GTO 16
496 RTN
497♦LBL 20
498 RDN
499 X<>Y
500 RCL 06
501 +
502 X<>Y
503 RCL 25
504 -
505 R-P
506 X<>Y
507 RCL 13
508 +
509 X<>Y
510 P-R
511 X<>Y
512 RCL 06
513 -
514 X<>Y
515 RCL 26
516 RCL 05
517 -
518 +
519 R-P
520 GTO 18
521 RTN
522♦LBL 22
523 RCL 17
524 -
525 X>Y?
526 GTO 26
527 GTO 23
528 RTN
529♦LBL 21
530 RCL 16
531 RCL 17
532 +
533 X>Y?
534 XEQ 22
535 X<>Y
536 RCL 16
537 RCL 17
538 2
539 /
540 +
541 ENTER↑
542 RCL 05
543 -
544 RCL 10
545 *
546 RCL 20
547 +
548 X<>Y
549 RCL 16
550 -
551 RCL 17
552 2
553 /
554 -
555 RCL 09
556 *
557 GTO 24
558 RTN
559♦LBL 23
560 -
561 ENTER↑
562 ENTER↑
563 RCL 09
564 RCL 10
565 -
566 1 E2
567 *
568 RCL 17
569 /
570 *
571 2
572 /
573 RCL 10
574 1 E2
575 *
576 +
577 *
578 1 E2
579 /
580 RCL 16
581 RCL 05
582 -
583 RCL 10
584 *
585 RCL 20
586 +
587 GTO 24
588 RTN
589♦LBL C
590 RCL 27
591 STO 20
592 GTO 10
593 RTN
594♦LBL F
595 RCL 28
596 STO 20
597 GTO 10
598 RTN
599♦LBL 00
600 RCL 19
601 RCL 18
602 RCL 26
603 -
604 R-P
605 X<>Y
606 RCL 13
607 -
608 CHS
609 X<>Y
610 P-R
611 RCL 25
612 +
613 STO 18
614 X<>Y
615 CHS
616 STO 19
617 X<>Y
618 ENTER↑
619 GTO 11
620 RTN
```

HP41-CV/CX
SOLUTIONS BOOKLETS

VERTICAL ALIGNMENT
Calculates CONTINUOUS vertical alignment without changing back and forth between Grade and Curve routines. Asymmetrical, Compound and Reverse Curves.

booklet #525 . . . . . **$6.95**
with programmed cards **$12.95**

SPIRAL CURVE SOLUTIONS
The only COMPLETE spiral curve program for use with hand-helds.

Booklet #529 . . . . . **$6.95**
with programmed cards **$12.95**

EDM SLOPE STAKING
Sets slope stakes and reference points from remote instrument and backsight locations directly.

Booklet #533 . . . . . **$6.95**
with programmed cards **$12.95**

TRIANGLE SOLUTIONS
The most complete triangle solutions program available. Solves with any of the following knowns: ASA SAA SAS SSA SSS Area-SS Area-AA Area-SA

Booklet #535 . . . . . **$6.95**
with programmed cards **$12.95**

POINT STORAGE CONVERSIONS
for the
HP41-CV/CX Surveying Pac

Traverse, Inverse & Sideshots
w/automatic compass correction.

Booklet #521 . . . . . **$6.95**
with programmed cards **$12.95**

INTERSECTION SOLUTIONS
(Auto-Inverse routine added)
Bearing-Bearing, Bearing-Dist, Dist-Dist and Offset to a Line.

Booklet #523 . . . . . **$6.95**
with programmed cards **$12.95**

Coordinate Transformation
Rotate or re-scale. Automatically outputs the new rotated bearings and distances between points.

Booklet #527 . . . . . **$6.95**
with programmed cards **$12.95**

PREDETERMINED AREAS
Calculate required areas from your pre-stored coordinate pairs.

Booklet #531 . . . . . **$6.95**
with programmed cards **$12.95**

Software by D'Zign
P.O. BOX 1370 • PACIFICA, CA 94044
[415] 355-8942

CA residents add 6.5% sales tax

COCO 41

This is the ROM you've been needing. Complete traverse, inverse . . . .everything the others have and more! Coordinate storage and retrieval option by **point number**. No need to pre-divide to mean angles before input in the field.

The closure routine includes **automatic angle check and adjustment**. The compass and transit corrections are automatic, with **auto-inverse** between the adjusted coordinates, all at the touch of a button. Does not require X-memory to work.

Contains keyboard functions for all of the common conversions; Feet to Meters, Feet to Foot-Inch-Fraction, Azimuth to Bearing, (option of North or South Azimuth), add or subtract feet-inch-fractions directly with one shifted keystroke! And, of course, inverse functions of all of these.

available early 1988

$6.95
41 887

ISBN 0-9616846-8-2